# NAG Fortran Library Routine Document

# F08QHF (DTRSYL)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F08QHF (DTRSYL) solves the real quasi-triangular Sylvester matrix equation.

## 2 Specification

```
      SUBROUTINE F08QHF (TRANA, TRANB, ISGN, M, N, A, LDA, B, LDB, C, LDC,
     1                   SCALE, INFO)
      INTEGER            ISGN, M, N, LDA, LDB, LDC, INFO
      double precision   A(LDA,*), B(LDB,*), C(LDC,*), SCALE
      CHARACTER*1        TRANA, TRANB
```

The routine may be called by its LAPACK name ***dtrsyl***.

## 3 Description

F08QHF (DTRSYL) solves the real Sylvester matrix equation

$$\mathrm{op}(A)X \pm X\,\mathrm{op}(B) = \alpha C,$$

where $\mathrm{op}(A) = A$ or $A^{\mathrm{T}}$, and the matrices $A$ and $B$ are upper quasi-triangular matrices in canonical Schur form (as returned by F08PEF (DHSEQR)); $\alpha$ is a scale factor ( $\leq 1$) determined by the routine to avoid overflow in $X$; $A$ is $m$ by $m$ and $B$ is $n$ by $n$ while the right-hand side matrix $C$ and the solution matrix $X$ are both $m$ by $n$. The matrix $X$ is obtained by a straightforward process of back-substitution (see Golub and Van Loan (1996)).

Note that the equation has a unique solution if and only if $\alpha_i \pm \beta_j \neq 0$, where $\{\alpha_i\}$ and $\{\beta_j\}$ are the eigenvalues of $A$ and $B$ respectively and the sign ($+$ or $-$) is the same as that used in the equation to be solved.

## 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (1992) Perturbation theory and backward error for $AX - XB = C$ *Numerical Analysis Report* University of Manchester

## 5 Parameters

1: TRANA – CHARACTER*1 *Input*

*On entry*: specifies the option $\mathrm{op}(A)$.

TRANA = 'N'

$\mathrm{op}(A) = A$.

TRANA = 'T' or 'C'

$\mathrm{op}(A) = A^{\mathrm{T}}$.

*Constraint*: TRANA = 'N', 'T' or 'C'.

2: TRANB – CHARACTER*1 *Input*

*On entry*: specifies the option op($B$).

TRANB = 'N'

op($B$) = $B$.

TRANB = 'T' or 'C'

op($B$) = $B^{\mathrm{T}}$.

*Constraint*: TRANB = 'N', 'T' or 'C'.

3: ISGN – INTEGER *Input*

*On entry*: indicates the form of the Sylvester equation.

ISGN = +1

The equation is of the form $\mathrm{op}(A)X + X\,\mathrm{op}(B) = \alpha C$.

ISGN = −1

The equation is of the form $\mathrm{op}(A)X - X\,\mathrm{op}(B) = \alpha C$.

*Constraint*: ISGN = +1 or −1.

4: M – INTEGER *Input*

*On entry*: $m$, the order of the matrix $A$, and the number of rows in the matrices $X$ and $C$.

*Constraint*: M ≥ 0.

5: N – INTEGER *Input*

*On entry*: $n$, the order of the matrix $B$, and the number of columns in the matrices $X$ and $C$.

*Constraint*: N ≥ 0.

6: A(LDA,∗) – ***double precision*** array *Input*

**Note**: the second dimension of the array A must be at least max(1, M).

*On entry*: the $m$ by $m$ upper quasi-triangular matrix $A$ in canonical Schur form, as returned by F08PEF (DHSEQR).

7: LDA – INTEGER *Input*

*On entry*: the first dimension of the array A as declared in the (sub)program from which F08QHF (DTRSYL) is called.

*Constraint*: LDA ≥ max(1, M).

8: B(LDB,∗) – ***double precision*** array *Input*

**Note**: the second dimension of the array B must be at least max(1, N).

*On entry*: the $n$ by $n$ upper quasi-triangular matrix $B$ in canonical Schur form, as returned by F08PEF (DHSEQR).

9: LDB – INTEGER *Input*

*On entry*: the first dimension of the array B as declared in the (sub)program from which F08QHF (DTRSYL) is called.

*Constraint*: LDB ≥ max(1, N).

10: C(LDC,$*$) – ***double precision*** array *Input/Output*

**Note**: the second dimension of the array C must be at least $\max(1, \mathrm{N})$.

*On entry*: the $m$ by $n$ right-hand side matrix $C$.

*On exit*: is overwritten by the solution matrix $X$.

11: LDC – INTEGER *Input*

*On entry*: the first dimension of the array C as declared in the (sub)program from which F08QHF (DTRSYL) is called.

*Constraint*: $\mathrm{LDC} \geq \max(1, \mathrm{M})$.

12: SCALE – ***double precision*** *Output*

*On exit*: the value of the scale factor $\alpha$.

13: INFO – INTEGER *Output*

*On exit*: $\mathrm{INFO} = 0$ unless the routine detects an error (see Section 6).

# 6 Error Indicators and Warnings

$\mathrm{INFO} < 0$

If $\mathrm{INFO} = -i$, the $i$th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\mathrm{INFO} = 1$

$A$ and $B$ have common or close eigenvalues, perturbed values of which were used to solve the equation.

# 7 Accuracy

Consider the equation $AX - XB = C$. (To apply the remarks to the equation $AX + XB = C$, simply replace $B$ by $-B$.)

Let $\tilde{X}$ be the computed solution and $R$ the residual matrix:

$$R = C - \left(A\tilde{X} - \tilde{X}B\right).$$

Then the residual is always small:

$$\|R\|_F = O(\epsilon)\left(\|A\|_F + \|B\|_F\right)\left\|\tilde{X}\right\|_F.$$

However, $\tilde{X}$ is **not** necessarily the exact solution of a slightly perturbed equation; in other words, the solution is not backwards stable.

For the forward error, the following bound holds:

$$\left\|\tilde{X} - X\right\|_F \leq \frac{\|R\|_F}{sep(A, B)}$$

but this may be a considerable overestimate. See Golub and Van Loan (1996) for a definition of $sep(A, B)$, and Higham (1992) for further details.

These remarks also apply to the solution of a general Sylvester equation, as described in Section 8.

# 8 Further Comments

The total number of floating-point operations is approximately $mn(m + n)$.

To solve the **general** real Sylvester equation

$$AX \pm XB = C$$

where $A$ and $B$ are general non-symmetric matrices, $A$ and $B$ must first be reduced to Schur form (by calling F08PAF (DGEES), for example):

$$A = Q_1 \tilde{A} Q_1^{\mathrm{T}}$$

and

$$B = Q_2 \tilde{B} Q_2^{\mathrm{T}}$$

where $\tilde{A}$ and $\tilde{B}$ are upper quasi-triangular and $Q_1$ and $Q_2$ are orthogonal. The original equation may then be transformed to:

$$\tilde{A}\tilde{X} \pm \tilde{X}\tilde{B} = \tilde{C}$$

where $\tilde{X} = Q_1^{\mathrm{T}} X Q_2$ and $\tilde{C} = Q_1^{\mathrm{T}} C Q_2$. $\tilde{C}$ may be computed by matrix multiplication; F08QHF (DTRSYL) may be used to solve the transformed equation; and the solution to the original equation can be obtained as $X = Q_1 \tilde{X} Q_2^{\mathrm{T}}$.

The complex analogue of this routine is F08QVF (ZTRSYL).

## 9 Example

This example solves the Sylvester equation $AX + XB = C$, where

$$A = \begin{pmatrix} 0.10 & 0.50 & 0.68 & -0.21 \\ -0.50 & 0.10 & -0.24 & 0.67 \\ 0.00 & 0.00 & 0.19 & -0.35 \\ 0.00 & 0.00 & 0.00 & -0.72 \end{pmatrix}, \quad B = \begin{pmatrix} -0.99 & -0.17 & 0.39 & 0.58 \\ 0.00 & 0.48 & -0.84 & -0.15 \\ 0.00 & 0.00 & 0.75 & 0.25 \\ 0.00 & 0.00 & -0.25 & 0.75 \end{pmatrix}$$

and

$$C = \begin{pmatrix} 0.63 & -0.56 & 0.08 & -0.23 \\ -0.45 & -0.31 & 0.27 & 1.21 \\ 0.20 & -0.35 & 0.41 & 0.84 \\ 0.49 & -0.05 & -0.52 & -0.08 \end{pmatrix}.$$

### 9.1 Program Text

```
*     F08QHF Example Program Text
*     Mark 16 Release. NAG Copyright 1992.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, LDA, LDB, LDC
      PARAMETER        (MMAX=8,NMAX=8,LDA=MMAX,LDB=NMAX,LDC=MMAX)
*     .. Local Scalars ..
      DOUBLE PRECISION SCALE
      INTEGER          I, IFAIL, INFO, J, M, N
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,MMAX), B(LDB,NMAX), C(LDC,NMAX)
*     .. External Subroutines ..
      EXTERNAL         DTRSYL, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08QHF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      IF (M.LE.MMAX .AND. N.LE.NMAX) THEN
*
*        Read A, B and C from data file
*
         READ (NIN,*) ((A(I,J),J=1,M),I=1,M)
         READ (NIN,*) ((B(I,J),J=1,N),I=1,N)
```

```
         READ (NIN,*) ((C(I,J),J=1,N),I=1,M)
*
*        Solve the Sylvester equation A*X + X*B = C for X
*
         CALL DTRSYL('No transpose','No transpose',1,M,N,A,LDA,B,LDB,C,
     +               LDC,SCALE,INFO)
         IF (INFO.EQ.1) THEN
            WRITE (NOUT,99999)
            WRITE (NOUT,*)
         END IF
*
*        Print the solution matrix X
*
         IFAIL = 0
         CALL X04CAF('General',' ',M,N,C,LDC,'Solution matrix X',IFAIL)
*
         WRITE (NOUT,*)
         WRITE (NOUT,99998) 'SCALE = ', SCALE
      ELSE
         WRITE (NOUT,*) 'MMAX and/or NMAX too small'
      END IF
      STOP
*
99999 FORMAT (/' A and B have common or very close eigenvalues.',/' Pe',
     +       'rturbed values were used to solve the equations')
99998 FORMAT (1X,A,1P,E10.2)
      END
```

## 9.2 Program Data

```
F08QHF Example Program Data
  4  4                             :Values of M and N
  0.10   0.50   0.68  -0.21
 -0.50   0.10  -0.24   0.67
  0.00   0.00   0.19  -0.35
  0.00   0.00   0.00  -0.72   :End of matrix A
 -0.99  -0.17   0.39   0.58
  0.00   0.48  -0.84  -0.15
  0.00   0.00   0.75   0.25
  0.00   0.00  -0.25   0.75   :End of matrix B
  0.63  -0.56   0.08  -0.23
 -0.45  -0.31   0.27   1.21
  0.20  -0.35   0.41   0.84
  0.49  -0.05  -0.52  -0.08   :End of matrix C
```

## 9.3 Program Results

```
 F08QHF Example Program Results

 Solution matrix X
          1       2       3       4
 1  -0.4209  0.1764  0.2438 -0.9577
 2   0.5600 -0.8337 -0.7221  0.5386
 3  -0.1246 -0.3392  0.6221  0.8691
 4  -0.2865  0.4113  0.5535  0.3174

 SCALE =   1.00E+00
```